APPENDIX 6. COLLISION FREQUENCY MODELS WITH INPUT PARAMETER FORMS

The following collision frequency models are available. The input parameter forms, which describe the model, and the subroutine listings are given on the pages shown.

To add other collision frequency models the user must write a subroutine that will calculate the normalized collision frequency (Z) and its gradient ($\partial Z/\partial r$, $\partial Z/\partial \theta$, $\partial Z/\partial \varphi$) as a function of position in spherical polar coordinates (r, θ, φ). ($Z = \nu/2\pi f$, where ν is the collision frequency between electrons and neutral air molecules and f is the wave frequency. If the Sen-Wyller formula for refractive index is used, then $Z = \nu_m/2\pi f$, where ν_m is the mean collision frequency.)

The restrictions on electron density models also apply to collision frequency models. The coordinates r, θ, φ refer to the computational coordinates system, which may not be the same as geographic coordinates. In particular, they are geomagnetic coordinates when the earth-centered dipole model of the earth's magnetic field is used.

The input to the subroutine (r, θ, ϕ) is through blank common. (See Table 3.) The output is through common block /ZZ/. (See Table 10.) It is useful if the name of the subroutine suggests the model to which it corresponds. It should have an entry point COLFRZ so that other subroutines in the program can call it. Any parameter needed by the subroutine should be input into W251 through W299 of the W array. (See Table 2.) If the model needs massive amounts of data, these should be read in by the subroutine following the example of subroutine TABLEZ.

INPUT PARAMETER FORM FOR SUBROUTINE TABLEZ

IONOSPHERIC COLLISION FREQUENCY PROFILE

The first card tells how many profile points in I4 format. The cards following the first card give the height and collision frequency of the profile points one point per card in F8.2, E12.4 format. The heights must be in increasing order. Set W250 = 1.0 to read in a new profile. After the cards are read, TABLEZ will reset W250 = 0.0. This subroutine makes an exponential extrapolation down using the bottom 2 points in the profile.

1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20	1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20
HEIGHT h km	COLLISION FREQUENCY ν COLLISIONS/sec.	HEIGHT h km	COLLISION FREQUENCY ν COLLISIONS/sec.

```
      SUBROUTINE TABLEZ
C     CALCULATES COLLISION FREQUENCY AND ITS GRADIENT FROM PROFILES
C     HAVING THE SAME FORM AS THOSE USED BY CROFTS RAY TRACING PROGRAM
C MAKES AN EXPONENTIAL EXTRAPOLATION DOWN USING THE BOTTOM TWO POINTS
C          NEEDS SUBROUTINE GAUSEL
      DIMENSION HPC(100),FN2C(100),ALPHA(100),BETA(100),
     1 GAMMA(100),DELTA(100),MAT(4,5),SLOPE(100)
      COMMON /CONST/ PI,PIT2,PID2,DJM(5)
      COMMON /ZZ/ MODZ,Z,PZPR,PZPTH,PZPPH
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(READNU,W(250))
      REAL MAT
      DATA (MODZ=6HTABLEZ)
      ENTRY COLFRZ
      IF (.NOT.READNU) GO TO 10
      READNU=0.
      READ 2, NOC,(HPC(I),FN2C(I),I=1,NOC)
    2 FORMAT(I4/(F8.2,E12.4))
      PRINT 1200, (HPC(I),FN2C(I), I=1,NOC)
 1200 FORMAT(1H1,14X,6HHEIGHT,4X,20HCOLLISION FREQUENCY
     1(1X,F20.10,E20.10))
      A=0.
      IF(FN2C(1).NE.0.) A=ALOG(FN2C(2)/FN2C(1))/(HPC(2)-HPC(1))
      FN2C(1)=FN2C(1)/PIT2*1.E-6
      FN2C(2)=FN2C(2)/PIT2*1.E-6
      SLOPE(1)=A*FN2C(1)
      SLOPE(NOC)=0.
      DO 5 I=2,NOC
      IF(I.EQ.NOC) GO TO 6
      FN2C(I+1)= FN2C(I+1)/PIT2*1.E-6
      DO 3 J=1,3
      M=I+J-2
      MAT(J,1)=1.
      MAT(J,2)=HPC(M)
      MAT(J,3)=HPC(M)**2
    3 MAT(J,4)=FN2C(M)
      CALL GAUSEL (MAT,4,3,4,NRANK)
      IF (NRANK.LT.3) GO TO 20
      SLOPE(I)=MAT(2,4)+2.*MAT(3,4)*HPC(I)
    6 CONTINUE
      DO 4 J=1,2
      M=I+J-2
      MAT(J,1)=1.
      MAT(J,2)=HPC(M)
      MAT(J,3)=HPC(M)**2
      MAT(J,4)=HPC(M)**3
      MAT(J,5)=FN2C(M)
      L=J+2
      MAT(L,1)=0.
      MAT(L,2)=1.
      MAT(L,3)=2.*HPC(M)
      MAT(L,4)=3.*HPC(M)**2
    4 MAT(L,5)=SLOPE(M)
      CALL GAUSEL (MAT,4,4,5,NRANK)
      IF (NRANK.LT.4) GO TO 20
      ALPHA(I)=MAT(1,5)
      BETA(I)=MAT(2,5)
      GAMMA(I)=MAT(3,5)
    5 DELTA(I)=MAT(4,5)
      JUP=2
   10 H=R(1)-EARTHR
      IF (H.GE.HPC(1)) GO TO 12
   11 JUP=2
      Z=FN2C(1)*EXP(A*(H-HPC(1)))/F
      PZPR=A*Z
```

```
      RETURN
   12 IF (H.GE.HPC(NOC)) GO TO 18
      NSTEP=1
      IF (H.LT.HPC(JUP-1)) NSTEP=-1
   15 IF (HPC(JUP-1).GT.H.OR.H.GE.HPC(JUP)) GO TO 16
      Z=(ALPHA(JUP)+H*(BETA(JUP)+H*(GAMMA(JUP)+H*DELTA(JUP))))/F
      PZPR=(BETA(JUP)+H*(2.*GAMMA(JUP)+H*3.*DELTA(JUP)))/F
      RETURN
   16 JUP=JUP+NSTEP
      IF (JUP.LT.2) GO TO 11
      IF (JUP.LT.NOC) GO TO 15
   18 JUP=NOC
      Z=FN2C(NOC)/F
      PZPR=0.
      RETURN
   20 PRINT 21, I,HPC(I)
   21 FORMAT(4H THE,I4,58HTH POINT IN THE COLLISION FREQUENCY PROFILE HA
     1S THE HEIGHT,F8.2,40H KM, WHICH IS THE SAME AS ANOTHER POINT.)
      CALL EXIT
          END
```

INPUT PARAMETER FORM FOR SUBROUTINE CONSTZ

An ionospheric collision frequency model consisting of a constant collision frequency

$\nu = 0$ for $h \leq h_{min}$

$\nu = \nu_0$ for $h > h_{min}$

Specify:

ν_0 = __________ collisions per second (W251)

h_{min} = __________ km (W252)

```
      SUBROUTINE CONSTZ                                                 CONZ001
C     CONSTANT COLLISION FREQUENCY                                      CONZ002
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)                                CONZ003
      COMMON /ZZ/ MODZ,Z,PZPR,PZPTH,PZPPH                               CONZ004
      COMMON R(6) /WW/ ID(10),WO,W(400)                                 CONZ005
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(NU,W(251)),(HMIN,W(252))      CONZ006
      REAL NU                                                           CONZ007
      DATA (MODZ=6HCONSTZ)                                              CONZ008
      ENTRY COLFRZ                                                      CONZ009
      H=R(1)-EARTHR                                                     CONZ010
      Z=0.                                                              CONZ011
      IF (H.GT.HMIN) Z=NU/(PIT2*F)*1.E-6                                CONZ012
      RETURN                                                            CONZ013
         END                                                            CONZ014-
```

INPUT PARAMETER FORM FOR SUBROUTINE EXPZ

An ionospheric collision frequency model consisting of an exponential profile

$$\nu = \nu_o e^{-a(h-h_o)}$$

h is the height above the ground

Specify:

The collision frequency at the height h_o, ν_o = ____________________ collisions per second (W251)

The reference height, h_o = ______________ km (W252)

The exponential decrease of ν with height, a = ______________ km^{-1} (W253)

```
      SUBROUTINE EXPZ
C        EXPONENTIAL COLLISION FREQUENCY MODEL
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /ZZ/ MODZ,Z,PZPR,PZPTH,PZPPH
      COMMON R(6) /WW/ ID(10),WO,W(400)
      REAL NU,NUO
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(NUO,W(251)),(HO,W(252)),
     1 (A,W(253))
      DATA (MODZ=6H EXPZ )
      ENTRY COLFRZ
      H=R(1)-EARTHR
      NU=NUO/EXP (A*(H-HO))
      Z=NU/(PIT2*F*1.E6)
      PZPR    =-A*Z
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE EXPZ2

An ionospheric collision frequency model consisting of a combination of two exponential profiles

$$\nu = \nu_1 e^{-a_1(h-h_1)} + \nu_2 e^{-a_2(h-h_2)}$$

where h is the height above the ground.

Specify for the first exponential:

Collision frequency at height h_1, ν_1 = ____________________ collisions per second (W251)

Reference height, h_1 = ______________ km (W252)

Exponential decrease of ν with height, a_1 = ___________ km^{-1} (W253)

Specify for the second exponential:

Collision frequency at height h_2, ν_2 = ____________________ collisions per second (W254)

Reference height, h_2 = ______________ km (W255)

Exponential decrease of ν with height, a_2 = ___________ km^{-1} (W256)

```
      SUBROUTINE EXPZ2                                                  XPZ2001
C          COLLISION FREQUENCY PROFILE FROM TWO EXPONENTIALS            XPZ2002
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)                                XPZ2003
      COMMON /ZZ/ MODZ,Z,PZPR,PZPTH,PZPPH                               XPZ2004
      COMMON R(6) /WW/ ID(10),W0,W(400)                                 XPZ2005
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(NU1,W(251)),(H1,W(252)),      XPZ2006
     1 (A1,W(253)),(NU2,W(254)),(H2,W(255)),(A2,W(256))                 XPZ2007
      REAL NU1,NU2                                                      XPZ2008
      DATA (MODZ=6H EXPZ2)                                              XPZ2009
      ENTRY COLFRZ                                                      XPZ2010
      H=R(1)-EARTHR                                                     XPZ2011
      EXP1= NU1* EXP(-A1*(H-H1))                                        XPZ2012
      EXP2= NU2* EXP(-A2*(H-H2))                                        XPZ2013
      Z=(EXP1+EXP2)/(PIT2*F*1.E6)                                       XPZ2014
      PZPR=(-A1*EXP1-A2*EXP2)/(PIT2*F*1.E6)                             XPZ2015
      RETURN                                                            XPZ2016
         END                                                            XPZ2017-
```